

Επιχειρησιακή Έρευνα
Ενότητα 2 - Λύσεις Ασκήσεων
Βελτιστοποίηση με ΗΥ - AMPL

Άσκηση 1

Σύνολα.

Το σύνολο των διαδρομών, Δ , με στοιχεία: Αθήνα-Θεσ ($j = 1$), Θεσ-Κών ($j = 2$), Αθήνα-Κών ($j = 3$).

Το σύνολο των πτήσεων, Π , με στοιχεία: Αθήνα-Θεσ ($k = 1$), Θεσ-Κών ($k = 2$).

Το σύνολο των τύπων των εισιτηρίων, T , με στοιχεία: Υ ($i = 1$), Β ($k = 2$), Μ ($k = 3$).

Παράμετροι.

c : η χωρητικότητα του αεροπλάνου,

p_{ij} : η τιμή του εισιτηρίου τύπου i για τη διαδρομή j , $i \in T, j \in \Delta$,

m_{ij} : μέγιστος αριθμός εισιτηρίων τύπου i για τη διαδρομή j , $i \in T, j \in \Delta$,

$d_{kj} = \begin{cases} 1 & , \text{αν η πτήση } k \text{ περιλαμβάνεται στη διαδρομή } j \\ 0 & , \text{αλλιώς} \end{cases}$, $k \in \Pi, j \in \Delta$.

Μεταβλητές απόφασης.

x_{ij} : αριθμός εισιτηρίων τύπου i για τη διαδρομή j , $i \in T, j \in \Delta$.

Το π.γ.π. είναι

$$\begin{aligned} \max & \quad \sum_{i \in T} \sum_{j \in \Delta} p_{ij} x_{ij} \\ \text{υπό} & \quad \sum_{i \in T} \sum_{j \in \Delta} d_{kj} x_{ij} \leq c \quad , k \in \Pi, \\ & \quad 0 \leq x_{ij} \leq m_{ij} \quad , i \in T, j \in \Delta. \end{aligned}$$

Τα δεδομένα είναι

$c = 300$,

τα p_{ij} δίνονται από τον πίνακα

$i \setminus j$	1	2	3
1	300	160	360
2	220	130	280
3	100	80	140

τα m_{ij} δίνονται από τον πίνακα

$i \setminus j$	1	2	3
1	40	80	30
2	80	130	100
3	220	200	180

και τα d_{kj} δίνονται από τον πίνακα

$k \setminus j$	1	2	3
1	1	0	1
2	0	1	0

Το αρχείο

Airticket_Scheduling.mod

με το μοντέλο που θα δώσουμε στην AMPL είναι:

```
##Airticket scheduling - Model

set Routes;
set Flights;
set Ticket_types;

param capacity;
param price{i in Ticket_types, j in Routes};
param max_tickets{i in Ticket_types, j in Routes};
param route_flight{k in Flights, j in Routes};

var number_tickets{i in Ticket_types, j in Routes};

maximize profit: sum{i in Ticket_types, j in Routes}
  price[i,j]*number_tickets[i,j];

subject to Capacity_flight{k in Flights}: sum{i in Ticket_types, j in Routes}
  route_flight[k,j]*number_tickets[i,j]<=capacity;
subject to Max_tickets{i in Ticket_types, j in Routes}:
  0 <=number_tickets[i,j]<=max_tickets[i,j];
```

Το αρχείο

Airticket_Scheduling.dat

με τα δεδομένα που θα δώσουμε στην AMPL είναι:

```
##Airticket scheduling - Data

set Routes:= ATH_SKG SKG_IST ATH_IST;
set Flights:= FL_ATH_SKG FL_SKG_IST;
set Ticket_types:= Y B M;

param capacity:=300;
param price: ATH_SKG SKG_IST ATH_IST:=
  Y      300    160    360
  B      220    130    280
  M      100     80    140;
param max_tickets: ATH_SKG SKG_IST ATH_IST:=
  Y      40     80     30
  B      80    130    100
  M     220    200    180;
param route_flight: ATH_SKG SKG_IST ATH_IST:=
FL_ATH_SKG    1     0     1
FL_SKG_IST    0     1     1;
```

Οι εντολές που θα δώσουμε στην AMPL είναι

```
reset;  
option solver cplex;  
model Airticket_Scheduling.mod;  
data Airticket_Scheduling.dat;  
solve;  
display number_tickets;
```

Άσκηση 2

Σύνολα.

Το σύνολο των εκπομπόμενων μέσων, E , με στοιχεία: τηλεόραση ($i = 1$), ραδιόφωνο ($i = 2$).

Το σύνολο των έντυπων μέσων, T , με στοιχεία: περιοδικά ($i = 3$), εφημερίδες ($i = 4$).

Το σύνολο των όλων των μέσων, M , $M = E \cup T$.

Παράμετροι.

c_i : το κόστος μιας διαφήμισης στο μέσο i , $i \in M$,

e_i : η αποτελεσματικότητα μιας διαφήμισης στο μέσο i , $i \in M$,

M_i : ο μέγιστος αριθμός επιτρεπόμενων διαφημίσεων στο μέσο i , $i \in M$,

m_i : ο ελάχιστος αριθμός επιτρεπόμενων διαφημίσεων στο μέσο i , $i \in M$,

c : το ποσό που μπορεί να δαπανηθεί σε διαφημίσεις,

p : το μέγιστο ποσοστό διαφημίσεων σε έντυπα μέσα επί του αριθμού εκπομπόμενων διαφημίσεων.

Μεταβλητές απόφασης.

x_i : αριθμός διαφημίσεων στο μέσο i , $i \in M$.

Το π.γ.π. είναι

$$\begin{array}{l} \max \quad \sum_{i \in M} e_i x_i \\ \text{υπό} \quad \sum_{i \in M} c_i x_i \leq c, \\ \quad \quad m_i \leq x_i \leq M_i, \quad i \in M, \\ \quad \quad \sum_{i \in T} x_i \leq p \sum_{i \in E} x_i. \end{array}$$

Δεδομένα.

Τα c_i και e_i δίνονται από τον πίνακα

i	1	2	3	4
c_i	1000	100	600	400
e_i	100	22	70	50

$M_3 = 12$ και $M_1 = M_2 = M = 4 = 1000$ (στην πραγματικότητα δεν περιορίζουν),

$m_1 = m_2 = 10$ και $m_3 = m_4 = 0$,

$c = 30000$, $p = 0.4$.

Το αρχείο

Advertisement_Campaign.mod

με το μοντέλο που θα δώσουμε στην AMPL είναι:

```
##Advertisement campaign - Model

set Print_Media;
set Air_Media;
set Media=Air_Media union Print_Media;

param cost{i in Media};
param efficiency{i in Media};
param max_advert{i in Media};
param min_advert{i in Media};
param max_cost;
param max_rate_print_air;

var number_advert{i in Media};

maximize Efficiency: sum{i in Media} efficiency[i]*number_advert[i];

subject to Max_cost:
    sum{i in Media} cost[i]*number_advert[i]<=max_cost;
subject to Max_advert{i in Media}:
    number_advert[i]<=max_advert[i];
subject to Max_rate_print_air:
    sum{i in Print_Media} number_advert[i]<=
        max_rate_print_air*(sum{i in Air_Media} number_advert[i]);
subject to Min_advert{i in Media}:
    number_advert[i]>=min_advert[i];
```

Το αρχείο

Advertisement_Campaign.dat

με τα δεδομένα που θα δώσουμε στην AMPL είναι:

```
##Advertisement campaign - Data

set Print_Media:= Magazine Journal;
set Air_Media:= TV Radio;

param:      cost      efficiency      max_advert      min_advert:=
    TV       1000      100              1000            10
    Radio    100       22              1000            10
    Magazine 600       70              12              0
    Journal  400       50              1000            0;

param max_cost:= 30000;
param max_rate_print_air:=0.40; 1      1;
```

Οι εντολές που θα δώσουμε στην AMPL είναι

```
reset;  
option solver cplex;  
model Advertisement_Campaign.mod;  
data Advertisement_Campaign.dat;  
solve;  
display number_advert;
```